KIT
de sobrevivencia em
CALCULO
Prof. Doherty Andrade - DMA- UEM
doherty@gauss.dma.uem.br

Multiplicadores de Lagrange

O Problem é: Maximizar a função $3x^2y$ para pontos (x,y) do circulo $x^2+y^2=1$:

Defina:

```
> restart:
> f:=3*x^2*y;
```

$$f := 3x^2y$$

```
> c:=x^2+y^2-1;
```

$$c := x^2+y^2-1$$

Para ver a geometria da situação, nós vamos plotar as superfícies $z=x^2y$ e $x^2+y^2=1$

- vamos procurar para o ponto mais alto da interseção :

```
> with(plots,display3d);
```

$$[display3d]$$

```
> setoptions3d(shading=ZHUE,style=patch);
```

$$\mathrm{setoptions3d}(shading=ZHUE, style=patch)$$

```
> F:=plot3d(3*x^2*y,x=-2..2,y=-2..2):
> G:=plot3d([cos(t),sin(t),u],t=0..2*Pi,u=-5..5,style=PATCH):
> display3d({F,G},view=-4..4);
```



O método dos multiplicadores de Lagrange diz que devemos encontrar os pontos criticos de:

> **g:=f-lambda*c;**

$$g := 3x^2y - \lambda(x^2 + y^2 - 1)$$

> **gx:=diff(g,x); gy:=diff(g,y); gl:=diff(g,lambda);**

$$gx := 6xy - 2\lambda x$$

$$gy := 3x^2 - 2\lambda y$$

$$gl := -x^2 - y^2 + 1$$

> **crits:=solve({gx=0,gy=0,gl=0},{x,y,lambda});**

$$crits := \{y = 1, x = 0, \lambda = 0\}, \{y = -1, x = 0, \lambda = 0\},$$
$$\{y = \mathrm{RootOf}(3\_Z^2 - 1), x = \mathrm{RootOf}(3\_Z^2 - 2), \lambda = 3\,\mathrm{RootOf}(3\_Z^2 - 1)\}$$

Os "RootOf"s que aparece na resposta é que os polinomios são quadráticos, tem duas soluções escondidas. Podemos ver seus valores usando o comando do Maple "allvalues":

> **crits:=crits[1],crits[2],allvalues(crits[3]);**

$$crits := \{y = 1, x = 0, \lambda = 0\}, \{y = -1, x = 0, \lambda = 0\}, \{x = \frac{1}{3}\sqrt{6}, \lambda = \sqrt{3}, y = \frac{1}{3}\sqrt{3}\},$$
$$\{x = -\frac{1}{3}\sqrt{6}, \lambda = \sqrt{3}, y = \frac{1}{3}\sqrt{3}\}, \{x = \frac{1}{3}\sqrt{6}, \lambda = -\sqrt{3}, y = -\frac{1}{3}\sqrt{3}\}, \{x = -\frac{1}{3}\sqrt{6}, \lambda = -\sqrt{3}, y = -\frac{1}{3}\sqrt{}$$

```
> crits[1];
```

$$\{y = 1, x = 0, \lambda = 0\}$$

```
> crits[2];
```

$$\{y = -1, x = 0, \lambda = 0\}$$

Existem seis pontos criticos -- vamos avaliar f em cada um deles:

```
> for i from 1 to 6 do subs(crits[i],f) od;
```

$$0$$

$$0$$

$$\frac{2}{3}\sqrt{3}$$

$$\frac{2}{3}\sqrt{3}$$

$$-\frac{2}{3}\sqrt{3}$$

$$-\frac{2}{3}\sqrt{3}$$

O maximo ocorre no terceiro e no quarto pontos criticos:

```
> crits[3]; crits[4];
```

$$\{x = \frac{1}{3}\sqrt{6}, \lambda = \sqrt{3}, y = \frac{1}{3}\sqrt{3}\}$$

$$\{x = -\frac{1}{3}\sqrt{6}, \lambda = \sqrt{3}, y = \frac{1}{3}\sqrt{3}\}$$

```
>
```